Lab 14 - Excel Lab Exercise
Tracing the Motion of a Projectile

01/24/2005

Name ____ Box / Period: _________

Due Date: _________

Introduction: Speadsheets are calculating devices that can repeat calculations with large amounts of data. They are very useful in physics, where large amounts of data can be easily inserted into a spreadsheet so the computer can do the calculations. In this exercise, you will calculate the path of a projectile moving in two dimensions using a spreadsheet.

Save this spreadsheet for future use. It is a very good trajectory calculator that you can use for solving almost any trajectory problem. After you finish constructing this spreadsheet you will be required to use it for answering questions about a trajectory in a WebAssignment.

Basics:
Each square of a spreadsheet is called a **Cell**. The location of a cell [its address] is given with a column letter and row number, e.g. B11, G3 and Q91 are typical cell addresses.

A cell can hold a **Text** string, a numerical **Value** (data), or a mathematical **Formula** (which also computes to a numerical value in all the situations of interest to us).

Text and **Value**s are just what you expect them to be. Enter the letters and numbers just as you would in any document file.

A **Formula** in a cell must begin with an equal sign (=). It can contain numbers, constants [e.g., **pi()**] and, most importantly, references to values contained in other cells. For example, a formula in cell location D6 might contain =2*C4; which means cell D6 will display the numeric contents of cell C4 multiplied by 2.

We will be using the following mathematical operators in our formulae. These are the same as those used in a typical calculator:

		Examples
Addition	+	=B3+N19
Subtraction	–	=A7–D6
Multiplication	*	=D16*B3
Division	/	=A5/B3
Exponentiation	^	=A17^2

Excel can also handle many more sophisticated functions. These include many mathematical, statistical and logical functions. Most important to us are the trigonometric functions; like sine, cosine and tangent. When using the trigonometric functions remember that all angles are assumed to be in radians. If you want to enter angular data in degrees, then you must convert it to radians before using any trigonometric function. Out put from the arc-functions is also in radians. Again, you will have to convert the radians to degrees.

Sine	sin(expression)	=sin(B3*pi()/180)
Cosine	cos(expression)	=cos(1-G11^3)
Tangent	tan(expression)	=tan(30*pi()/180)
Arcsine	asin(expression)	=asin(D12)*180/pi()
Arccosine	acos(expression(	=acos(sqrt(3)/2)*180/pi()
Arctangent	atan(expression)	=atan(H5/P14)

To convert degrees to radians, multiply the angle in degrees by

pi()/180

Similarly, to convert radians to degrees, multiply the angle in radians by

180/pi()

In a spreadsheet it is often necessary to use parentheses liberally to avoid ambiguities in defining formulae. This is especially true whenever a complex expression appears in a numerator or a denominator of an expression expressed as a ratio.

Filling-Down or copying a formula reproduces it in new cells where the copies can operate on fresh data. Click with the mouse on a formula cell and then drag down with the mouse to highlight cells in the column below the original equation. Go to the Edit menu and select Fill-Down. The formula will be copied into all the highlighted cells. Each copy of the formula uses data in surrounding cells at the same relative positions to its new address.

When copying a formula, you can force the computer to use the data from a specific cell no matter where you copy the formula. This requires use of an **Absolute Cell Reference** within the formula. Use dollar signs ($) in a cell reference to force the formula to continue using that cell in calculations after being copied to a new location. For example, a formula that reads: =F2*$B$6 will always use the number in cell B6 no matter where you copy the formula.

The following section describes the equations you will use to model the path of a projectile. Copy all formulae in the main table down to row 45 to complete that table. Before you start the table, type in the various text headings as they are shown below and on the last page of this handout. The upper section of your spreadsheet contains the variables that you must enter by hand. These are the numbers that your formulas will access using absolute cell addresses. The table in the lower section contains most of the calculations.

Tracing the Path of a Projectile in Two Dimensions:

1. Start Excel and open a new spreadsheet.
2. Type in the data labels shown below. Enter the appropriate units in column C, across from each label in column A. Adjust the width of column A to hold all the labels shown.



3. Type 15 into B1, 30.00 into B2, 70.00 into B3, –9.81 into B4, 0.10 into B5.	*15.0 m/s is the initial velocity;* *30 degrees is the launch angle;* *70 m is the initial altitude;* *–9.81 m/s^2 is the acceleration of gravity;* *0.1 s is the time increment between calculations.*

Before you do anything else, save the spreadsheet on your H: drive. Use a name like **Lab-14-Trajectory Calculator.xls**. Save it often as you continue to work on the spreadsheet.

4. Into B6 enter a formula to convert degrees in cell B2 to radians
   =B2*pi()/180
5. Into B7 enter a formula to find the i-component (horizontal) of the initial velocity.
   =B1*cos(B6) *(Note that trig functions must have angles in radians.)*
6. Into B8 enter a formula to find the j-component (vertical) of the initial velocity.
   =B1*sin(B6)

7. Into A11 enter the starting time. For now enter zero. This only determines when the calculations begin. The projectile is always launched at time t=0. To focus in on different parts of the trajectory you could change the starting time and the time interval for your calculations in order to examine any piece of the trajectory.
8. Into A12 enter a formula to get to the next calculation time. This must be the previous time plus the time increment. Use a relative address for the prior time (A11), and an absolute address for the time increment ($B$5). All the copies of this equation will add the same time interval to the time in the cell above to get the current time.

   =A11+$B$5
9. With the mouse, click on A12 and drag down to A45. Click on Edit/Fill/Down. Column A is now showing times beginning with zero and increasing by 0.1 sec. You now have a specific set of times for which calculations will be performed.
10. In B11, enter the formula:

    =$B$7

    Drag B11 down to B45 and Edit/Fill/Down. Since the horizontal speed never changes, the absolute cell reference insures that the same horizontal speed applies to all times following the launch.
11. In column C, starting at cell C11, we calculate the vertical speed at each time. The physics equation looks like: $V_y = V_{y_o} + gt$ or

    "Vert V" = "Vert V_0" + g*Time

    which in cell C11 should look like

    =$B$8 + $B$4*A11

    Drag C11 down to C45 and Edit/Fill/Down. Absolute cell references insure that the correct gravitational acceleration (from $B$4) is used at all times and that the corrected value for the initial vertical velocity (from $B$8) is used at all times.
12. Column D computes the speed of the projectile at each time step. It is the vector sum of the components given in columns B and C: Speed = $(V_x^2 + V_y^2)^{1/2}$:

    =(B11^2+C11^2)^0.5

    Drag D11 down to D45, then Edit/Fill Down.
13. The angle (in column E) is the angle, in radians, of the projectile's path at each time step. You can find this by taking the inverse tangent of V_y/V_x.

    =atan(C11/B11)

    Drag E11 down to E45, then Edit/Fill Down.
14. Convert the angle in radians in column E to degrees in column F. Enter the correct formula in cell F11, then Edit/Fill Down to F45.

    (***Don't skip this step. Do you know how to convert radians to degrees?***)

    Write the formula here: ____
15. In Column G the x-position is computed at time t using the equation: $x = V_x t$

    =B11*A11
16. The y-position will be given by the equation: $y = y_o + V_{y_o} t + \frac{1}{2}gt^2$

    =$B$3+$B$8*A11+0.5*$B$4*A11^2.

After it is approved by your instructor, save the spreadsheet again on your H:\ drive.

Date Headings and Column Headings.

Format the headings as shown here; use the underline for Name and Box#.

	A	B	C	D	E	F	G	H
1	**Muzzle Speed**		m/s		Name			
2	**Launch Angle**		degrees					
3	**Initial Altitude**		m				Box #	
4	**Accel of Gravity**		m/s^2					
5	**Time Interval**		s					
6	**Launch Angle**		radians					
7	**Vo-Horizontal**		m/s					
8	**Vo-Vertical**		m/s					
9								
10	**Time**	**Vx**	**Vy**	**Speed**	**Ang (rad)**	**Ang (deg)**	**X-position**	**Y-position**
11								

Answer these questions about this projectile:

1. How long is the projectile in the air? ____

2. At what angle is it traveling just before it hits ground level? ______________________

3. What is the maximum height reached? _______________________________________

4. What is the height at 1.00 sec into the flight? _________________________________

5. What is the speed at 1.00 sec into the flight? ________________________________

6. What is the acceleration at 1.00 sec into the flight? ____________________________